
Supplementary information

Extraction procedure of the schizanthines

Stem-bark of Schizanthus grahamii was collected in Central Chile in January 2000.  The raw material was extracted 
with ethanol at room temperature.  After filtration, the solvent was evaporated to dryness.  The residue was taken up in 
0.1 M HCl and washed with dichloromethane.  The aqueous solution was basified with ammonium hydroxide to pH 12 
and extracted with dichloromethane.  The organic solvent was evaporated yielding a gummy alkaline residue.  Further 
purification onto an aluminium oxide column was performed according to Muñoz et al. (Muñoz O, Piovano M, 
Garbarino J, Hellwig V, Breitmaier E. 1996 Phytochemistry 43: 709-713) leading to a purified fraction containing the 
investigated compounds.

The pulse program for Bruker DRX spectrometers

;bvhmbcetgpl2nd_CT_2 modified to be constant-time (d14) based on 
; hmbcetgpl2nd 
; HMBC
;2D H-1/X correlation via heteronuclear zero and double quantum
;   coherence
;phase sensitive using Echo/Antiecho gradient selection
;with two-fold low-pass J-filter to suppress one-bond correlations
;no decoupling during acquisition
;
;D.O. Cicero, G. Barbato & R. Bazzo, J. Magn. Reson. 148, 
;   209-213 (2001)
; For CT, see Furihata, K. & Seto, Tetrahedron Letters, 1998, 39, 7337-7340 

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA1 = { 1.000 -cnst30}
define list<gradient> EA2 = { -cnst30 1.000}

"p2=p1*2"

"d0=3u"

"d6=1s/(cnst13*2)"

;constant time initialization
"d14=in0*(td1/2)+4u"
"in14=in0"

"DELTA1=1s/(2 * cnst6)-p16-d16"
"DELTA2=1s/(2 * cnst7)-p16-d16"
"DELTA3=d6-p16-d16-4u"

"DELTA4=p2+d0*2"

1 ze 
2 d1
3 p1 ph1
  DELTA1 UNBLKGRAD
  p16:gp3
  d16 pl2:f2
  (p3 ph3):f2
  DELTA2
  p16:gp4
  d16
  (p3 ph3):f2
  4u
  p16:gp5
  d16
  DELTA3
  d14             ;added for CT
  p2 ph1          ;added for CT
  d14             ;added for CT
  (p3 ph4):f2
  d0
  p2 ph2
  d0
  p16:gp1*EA1
  d16
  (p4 ph5):f2
  DELTA4
  p16:gp1*EA2
  d16 pl2:f2
  (p3 ph5):f2
  4u BLKGRAD
  go=2 ph31
  d1 mc #0 to 2 
    F1EA(igrad EA1 & igrad EA2, id0 & dd14 & ip4*2 & ip31*2 )

exit 
  

ph1=0
ph2=0 0 0 0 2 2 2 2
ph3=0 0 2 2
ph4=0 2
ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph31=0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;sp7: f2 channel - shaped pulse (180degree refocussing)

;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse                         [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                           [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d14 : constant time
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d16: delay for homospoil/gradient recovery
;d27:
;cnst6: = 1J(XH)min
;cnst7: = 1J(XH)max
;cnst13: = J(XH) long range
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: echo-antiecho

;use gradient ratio:    gp 1 : gp 3 : gp 4 : gp 5
;                         80 :   15 :  -10 :   -5

;for z-only gradients:
;gpz1: 80%
;gpz3: 15%
;gpz4: -10%
;gpz5: -5%

Details of the NMR experiments

Spectra of the crude mixture of pine resin (ca. 100 mg in CDCl3) were recorded on a Bruker DRX 500 spectrometer 
equipped with a direct detection probe optimized for carbon acquisition. The pulse program ( invietgpsi ) is a phase-
sensitive HSQC using sensitivity improvement and Echo/Antiecho-TPPI gradient selection with decoupling during 
acquisition (A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn. Reson. 93, 151-170 (1991); L.E. Kay, P. 
Keifer & T. Saarinen, J. Am. Chem. Soc. 114, 10663-5 (1992); J. Schleucher, M. Schwendinger, M. Sattler, P.Schmidt, 
O. Schedletzky, S.J. Glaser, O.W. Sorensen & C. Griesinger, J. Biomol. NMR 4,  301-306 (1994))  The delays of the 
INEPT were set for 1JCH = 145 Hz and 4 scans were recorded per time increment. In order to have XX0.0 ppm at the 
spectral boundaries in 10 ppm spectra, we set the carier frequency to 115 ppm but any valuex XX5.0 also work.  

Spectra of schizanthines (ca. 1 mg dissolved in 700 ul CDCl3) were recorded on a Bruker 500 MHz spectrometer 
equipped with a low-temperature detection probe.  The pulse program above and 1J(XH)max and 1J(XH)min were set to 
100 and 180 Hz respectively while nJCH = 10 Hz.  The number of scan/increment was set to 32 for SNR reasons.  

Comment concerning the quadrature detection in the carbon dimension

When the quadrature mode in F1 is E/AI (today’s standard when using gradients quipped spectrometers) QF, States or 
States-TPPI signals are back-folded.  When using TPPI quadrature, signals are folding at the boundaries of the 
spectrum instead of being back-folded. The latter mode is therefore not appropriate for this application!

Comment concerning the processing of 10 ppm spectra

When applying zero filling and squared cosine function as apodization for the processing, the signal full line-width at 
half-height are 1 ppm in the full and 0.04 ppm in the 10 ppm spectra.  Less time increments, say 128, would only 
provide one common digit making it more difficult to pair the signals and provide carbon chemical shifts with only five 
decimal figures. Note that when signal to noise is not a problem, the combination of three spectra with 240, 20 and 2 
ppm width could be a good choice for automatic spectral reconstruction of three spectra with 256 time increments.

